

Université Montpellier-II
UFR des Sciences - Département Informatique

Objets Avancés

**Vue générale de Smalltalk :
Tout Objet, Environnement de Génie Logiciel, Réflexivité,
Métaprogrammation, IDM,**

Notes de cours - 2007-2021

Christophe Dony

1 Introduction

Smalltalk est à la fois (i) un langage premier sur différents aspects du génie logiciel dont l'adaptabilité dynamique des programmes et la pratique "agile", (ii) un modèle d'évolution pour les autres langages de par la simplicité de ses concepts et, (iii) avec Simula, un des deux langages historiques à l'origine de l'approche Objet. Ce sont trois raisons pour l'étudier.

1.1 Idées clé

1. Simplicité, Uniformité (4 grands fondements dont “tout est objet”, voir section 1.3)
2. Langage de programmation ET Environnement interactif de génie logiciel dédié au développement “agile” ()




Figure (1) – *Ce cours utilise le langage et environnement Pharo, une émanation actuelle de Smalltalk - <https://pharo.org/>*

3. Ingénierie Dirigée par les Modèles, modèles à l’exécution, adaptabilité dynamique des programmes ... (“*L’intelligence c’est la faculté de s’adapter au changement.*” *Stephen Hawking*).

1.2 Quelques liens historiques et pratiques

- Alan Kay (https://fr.wikipedia.org/wiki/Alan_Kay), créateur du langage, médaille Turing
- Histoire, l'idée de base : <https://fr.wikipedia.org/wiki/Dynabook>
- Histoire, les livres : <http://www.world.st/learn/books>
- La machine virtuelle pour le tout objet réflexif : CoG de Eliot Miranda (<http://www.mirandabanda.org/cogblog/about-cog/>)
- Le centre d'information Européen : <http://www.esug.org/>
- Le langage utilisé utilisé dans ces notes : <http://pharo.org/> (Pharo is a clean, innovative, open-source Smalltalk-inspired environment)
- Un cours interactif en ligne : <http://mooc.pharo.org>,
- <http://stephane.ducasse.free.fr/FreeBooks.html> : catalogue de livres sur Smalltalk dont certains en ligne
- www.squeak.org : Squeak est un Smalltalk gratuit réalisé par une équipe dirigée par Alan Kay
- les versions industrielles : <http://www.cincomsmalltalk.com/main/> : Cincom, Smalltalk industriel (ObjectStudio - Visualworks)

- POURQUOI c'est intéressant :

- Simplicité : <http://stackoverflow.com/questions/1821266/what-is-so-special-about-small>
- Adaptabilité Dynamique, Ingénierie dirigée par les modèles : <http://moosetechnology.org/>
- Agilité : https://en.wikipedia.org/wiki/Extreme_programming. site de Kent Beck, fondateur de “Agile development” : <http://www.threeriversinstitute.org> :
- Innovation : <https://blog.appacademy.io/pharo-future-software-development/>

1.3 Le style de langage

- héritier de *Simula* (Objet, abstraction de données, polymorphisme d'inclusion)
- héritier de *Lisp-Scheme*
 - style applicatif (toute instruction est une expression),
 - ordre supérieur
 - typage dynamique (pas de problèmes de redéfinitions covariantes, pas de surcharge statique),
 - passage par référence (à la *Lisp*) généralisé,
 - allocation et récupération dynamiques et automatiques de la mémoire,
 - compilation en instructions d'une machine virtuelle,
 - réflexivité : Programmes et Environnement = données,

Le langage en quatre idées

1. Toute entité est un objet.

Un objet est une entité individuelle, repérée par une adresse unique, possédant un ensemble de champs (autant que d'attributs sur la classe), connus par leurs noms et contenant une valeur qui peut changer (mutable).

2. Tout objet est instance d'une classe, qui définit sa structure (définie par un ensemble attributs privés) et ses comportements (définis par un ensemble de méthodes).
3. Tout comportement d'un objet (méthode) est activé par un envoi de message (avec liaison dynamique).
4. Toute classe (sauf `ProtoObject`) est définie comme une spécialisation d'une autre. La relation de spécialisation définit un arbre d'héritage.

1.4 Interprétation, Compilation, machine virtuelle

Principe d'exécution : compilation des instructions en *byteCodes* ou instructions d'une machine virtuelle dédiée à la programmation par objets, puis interprétation de ces instructions.

Machine virtuelle Smalltalk : ensemble d'instructions dédiées à la programmation par objets et interpréteur associé. *Java* a repris ce schéma d'exécution. Voir par exemple COG (<http://www.mirandabanda.org/cogblog/about-cog/>).

Exemple de méthode et de bytecode généré

```
1 fact
2   ^self = 0
3     ifTrue: [1]
4     ifFalse: [self * (self - 1) fact]
```

Listing (1) – définition de la méthode fact sur la classe PositiveInteger

```
1 normal CompiledMethod numArgs=0 numTemps=0 frameSize=12
2 literals: (#fact )
3 1 <44> push self
4 2 <49> push 0
5 3 <A6> send =
6 4 <EC 07> jump true 13
7 6 <44> push self
8 7 <44> push self
9 8 <4A> push 1
10 9 <A1> send -
11 10 <70> send fact
12 11 <A8> send *
13 12 <66> pop
14 13 <60> push self; return
```

1.5 Introduction pratique

Classes et instances

```
2 Object subclass: #Compteur
3   instanceVariableNames: 'valeur'
4   classVariableNames: ''
5   poolDictionaries: ''
6   category: 'ExosPharo-PetitsExercices'

8 initialize
9   valeur := 0.

10 get
11   ^valeur

12 incr
13   valeur := valeur + 1.

14 decr
15   valeur := valeur - 1.
```

Listing (2) – classe Compteur, définition de classe et de méthodes d'instance

```
2 new
3   "redéfinition de la méthode new pour appeler automatiquement une méthode
4     d'initialization"
5   ^super new initialize
6
7 example
8   | c |
9   c := Compteur new.
10  c incr.
11  ^c
```

Listing (3) – définition de méthodes de classe

1.6 La classe Compteur dans l'environnement

The screenshot displays the Pharo environment with the following components:

- Playground:** A window titled "Playground" containing the text "Compteur new".
- Inspector on a Compteur:** A window showing the state of an instance of the class. It has two tabs: "Raw" and "Meta". The "Raw" tab is active, showing a table with two columns: "Variable" and "Value".

Variable	Value
self	a Compteur
valeur	0
- Class Editor:** A window titled "Compteur" showing the class definition. The "Class" tab is active, displaying the following source code:

```
Object subclass: #Compteur
  instanceVariableNames: 'valeur'
  classVariableNames: ''
  package: 'ExosPharo-PetitsExercices'
```

Below the code, the text "Définition de la classe Compteur." is visible. To the right, a "History Navigator" shows a list of methods: "decr", "get", "incr", "initialize", "raz", and "set:". The "initialize - release" method is currently selected.

Figure (2) – Définition de la classe Compteur - Pharo Env.

1.7 L'exception universelle : `doesNotUnderstand`

```
1 3.4 factorial
```

→ *Instance of SmallFloat64 did not understand factorial*

Conduite à tenir :

- lire le message
- si non compréhension, ouvrir le débbugger (one click), corriger, relancer

1.8 Tout objet et réflexivité

```
1 1 class "-> SmallInteger"  
2 1 class class "->SmallInteger class"  
3 1 class class class "->Metaclass"  
4 1 class class class class "->Metaclass class"  
5 1 class class class class class "->Metaclass"
```

```
1 5 factorial "->120"  
2 m := Integer compiledMethodAt: #factorial.  
3 m class "->CompiledMethod"  
4 m name "->'Integer>>#factorial"  
5 m valueWithReceiver: 5 arguments: #() "->120"
```

```
1 factorial  
2   ^self = 0  
3     ifTrue: [thisContext inspect.  
4               1]  
5     ifFalse: [self * (self - 1) fact]
```

2 Syntaxe

`exampleWithNumber: x`

"A method that illustrates every part of Smalltalk method syntax except primitives. It has unary, binary, and keyboard messages, declares arguments and temporaries, accesses a global variable (but not an instance variable), uses literals (array, character, symbol, string, integer, float), uses the pseudo variables true, false, nil, self, and super, and has sequence, assignment, return and cascade. It has both zero argument and one argument blocks."

```
| y |
true & false not & (nil isNil) ifFalse: [self halt].
y := self size + super size.
#(\ $a #a "a" 1 1.0)
  do: [ :each |
      Transcript show: (each class name);
      show: ' ' ].
^x < y
```

Listing (4) – from : <http://wiki.c2.com/?SmalltalkSyntaxInaPostcard>

- <http://www.chimu.com/publications/JavaSmalltalkSyntax.html> : Une comparaison des syntaxes Java et Smalltalk.
- <http://www.csci.csusb.edu/dick/samples/smalltalk.syntax.html> : Une présentation concise de la syntaxe de Smalltalk.

2.1 Constantes littérales

“Constantes” car leur valeur ne peut être modifiée et “littérales” car elles peuvent être entrées littéralement dans le texte des programmes.

- nombres : 3, 3.45, -3,
- caractères : \$a, \$M, \$\$
- chaînes : 'abc', 'the smalltalk system'
- symboles : #bill, #a22
- booléens : true, false
- tableaux de constantes littérales (construit à l'analyse syntaxique) : #(3 3.45 -3 aM 'abc' #bill true #(am stram gram))
- commentaires : "ce commentaire sera ignoré par l'analyseur syntaxique"

2.2 Identificateur

- Suite de chiffres et lettres commençant par une lettre.
- Manipulables dans le texte des programmes via leur symbole éponyme.

```
Integer methodDictionary at: #factorial
```

- Non typés dans le texte des programmes (typage dynamique).

2.3 Affectation

```
i := 3
```

Listing (5) – à la Algol, ou à la Pascal

2.4 Séquence d’Instructions

Il y a une séquence implicite associée à tout bloc (corps de méthode par exemple). Dans un bloc, Les instructions sont séparées par un “.” (et pas un “;”).

```
uneMethode  
  i := 1.  
  j := 2.
```

2.5 Tableaux littéraux

```
a := 3.  
{ a . a+1 . a*a } ”--> #(3 4 9)
```

2.6 Retour à l'appelant

```
^33 "return 33"
```

Toute invocation de méthode rend une valeur, celle de **self** par défaut, le type **void** n'existe pas.

2.7 Envoi de message

Définitions :

- Conceptuelle : demande à un objet d'exécuter un de ses comportements implanté sous la forme d'une méthode.
- Opérationnelle : appel de méthode avec sélection de la méthode à exécuter selon le type dynamique du premier argument, qualifié de receveur et syntaxiquement distingué.

Trois sortes de messages (distinction syntaxique) :

1. *unaires* (à un seul argument : le receveur)

1 **class**

5 **fact**

2. *binaires* (à deux arguments, syntaxe pratique pour les opérations arithmétiques en infixé)

1 + 2

3. “*keywords*” (à $n(n \geq 2)$ arguments dont le receveur)

Leur forme syntaxique (originale) permet de représenter un envoi de message comme une “petite conversation” (*small talk* en anglais) entre un objet et un autre.

```
1 log: 10
```

```
anArray at: 2 put: 3
```

```
monCalendrier enregistre: unEvenement le: unJour à: uneheure
```

Les familiers de *Java* traduiront en :

```
monCalendrier.enregistre(unEvenement, unJour, uneHeure)
```

Noms des méthodes utilisées dans ces exemples :

```
log:,
```

```
at:put:,
```

```
enregistre:le:à:.
```

- Précédence :
unaire > binaire > keyword,
exemple : $1 + 5 \text{ fact}$ égale 121 et pas 720.
- Associativité : A précedence égale les messages sont composés de gauche à droite.
exemple :
 $2 + 3 * 5 = 25$
 $2 + (3 * 5) = 17$
- Cascade de messages :
 $r \ s1; \ s2; \ s3.$ est équivalent à $: r \ s1. \ r \ s2. \ r \ s3.$

2.8 Structures de contrôle

Il n'existe aucune forme syntaxique particulière pour les structures de contrôle ; le contrôle se réalise par envois de messages.

Le propre d'une structure de contrôle est d'avoir une politique non systématique d'évaluation de ses arguments. Implanter des structures de contrôle comme des méthodes standard nécessite donc que les arguments de ces méthodes ne soient pas évalués systématiquement au moment de l'appel. La solution Smalltalk à ce problème consiste à utiliser des *blocks* (fermetures lexicales), soit en position de receveur, soit en position d'argument.

Une fermeture lexicale est une fonction anonyme (lambda-expression) capturant son environnement lexical de définition (toute variable libre y est interprétée relativement à l'environnement dans lequel la fermeture a été définie).

La syntaxe de définition d'une fermeture est [<parametre>* | <instruction>*]

2.8.1 Exemples

Conditionnelle

```
(aNumber \% 2) = 0 ifTrue: [parity := 0] ifFalse: [parity := 1]
```

Boucle “For”

La boucle for utilise un *block* à un paramètre qui tient lieu de compteur de boucle

```
1 to: 20 do: [:i | Transcript show: i printString].
```

Itérateur généralisé

```
untableau := #(3 5 7 9).  
untableau do: [:each | Transcript show: each; cr]
```

Formes de gestion d'exceptions

```
untableau findKey: x ifAbsent: [0]
```

3 Pratique basique du langage

3.1 Classes et Méthodes

Toute classe est créée comme sous-classe d'une autre classe.

Variables d'instance, variables de classes (équivalent des “statiques” de C++ et Java), variables de *pool*.





Figure (3) – Définition de la classe TimeSpan - Pharo env.

Méthodes d'instance



The screenshot shows a software development environment window titled "Timespan>>#=". The window is divided into several panes. On the left, there is a "Scopes" pane showing a hierarchy of packages: Kernel, BasicObjects, Chronology, Classes, Copying, and Exceptions. The "Chronology" package is selected. In the center, there is a "Variables" pane showing a list of classes: ProtoObject, Object, Magnitude, Timespan (selected), Date, Month, and Schedule. On the right, there is a "History Navigator" pane showing a list of methods: -- all --, accessing, arithmetic (selected), conversion, enumerating, printing, private, and testing. Below the panes, there is a code editor showing the implementation of the "=" method for the Timespan class:

```
= comparand
  ^ self species = comparand species
  and: [ self start = comparand start
        and: [ self duration = comparand duration ] ]
```

Figure (4) – Méthode d'instance “=” de la classe TimeSpan

Méthodes de classe

Méthodes invocables par envoi de message aux classes ^a

```
Date new. Date today.
```

a. Les méthodes de classe sont en fait des méthodes d'instance définies sur les métaclasse, voir la section 4.3. En première approche, elles ressemblent aux *static* en C++ puis Java, mais elles sont en fait des méthodes standard définies sur les classes des classes (tout est objet), utilisables de façon standard.




Figure (5) – Méthode de classe “today” de la classe Date (En fait une méthode d’instance de la (méta)-classe “Date class”)

Différentes sortes de variables accessibles au sein des méthodes

Les identificateurs accessibles au sein d'une méthode M :

- les paramètres de M,
- les variables temporaires de M,
- les attributs (variables d'instance) déclarés sur la classe du receveur courant,
- les identificateurs prédéfinis `self`, `super`, `true`, `false`, `nil`.
- les variables de classe de la classe C de O (celle ou est définie M)
- les variables partagées entre la classe C et d'autres classes (voir `PoolDictionary`),
- les variables globales du système, dont celles référençant toutes les classes.

3.2 Envoi de message

Demande à un objet d'exécuter un de ses services.

Procédé, réalisé à l'exécution, menant à l'exécution d'une méthode de même nom que le *selecteur*.

Consiste en la recherche de la méthode dans la classe du receveur du message puis en cas d'échec dans ses superclasses.

Diverses techniques d'optimisation de l'envoi de messages sont implantées dans les diverses machines virtuelles dont la première est la technique de cache (voir le *green book*).

3.3 Instantiation et initialisation des objets

- On instancie une classe en lui envoyant le message `new`.
- `new` est une méthode définie sur une superclasse commune à toutes les métaclasses (cf. métaclasses).
`new` réalise l'allocation mémoire et rend la référence sur le nouvel objet.
- Exemple : `Date new, d := Date new.`

Il n'y a pas de constructeurs spécifiques ;

l'initialisation s'effectue via des méthodes standard, souvent invoquées automatiquement dans des redéfinitions de la méthode de classe `new`.

Exemples : voir les méthodes `new` et `x:y:` de la classe `Point`.

3.4 Exemple

```
1 Object subclass: #Compteur
2   instanceVariableNames: 'valeur'
3   classVariableNames: ''
4   poolDictionaries: ''
5   category: 'Compteur-MVC'

7 !Compteur methodsFor: 'acces lecture/ecriture'!
8 nombreUnites
9   ^valeur

11 nombreUnites: n
12   valeur := n.

14 !Compteur methodsFor: 'imprimer'!

16 printOn: aStream
17   aStream nextPut: $@.
18   self nombreUnites printOn: aStream
```

```
1 !Compteur methodsFor: 'operations'!  
  
3  decr  
4     self nombreUnites: valeur - 1  
  
6  incr  
7     self nombreUnites: valeur + 1  
  
9  raz  
10     self nombreUnites: 0  
11  ”-----”!  
  
13 Compteur class  
14     instanceVariableNames: ''  
  
16 !Compteur class methodsFor: 'creation'!  
17 new  
18     ^super new raz
```

3.5 Classe abstraite

Une classe ne peut être déclarée abstraite mais elle peut être rendue abstraite par le programmeur en redéfinissant la méthode `new` pour signaler une exception.

Ceci pose néanmoins de sérieux problèmes pour ses futures sous-classes concrètes. (utilisation obligatoire de `basicNew`). Ceci est une limite de la solution Smalltalk pour les métaclasse (voir section [4.3](#)).

3.6 Methode abstraite

Une méthode ne peut être déclarée abstraite mais elle peut être rendue abstraite en la définissant de la façon suivante :

```
method
  self subclassResponsibility
```

3.7 Polymorphisme d'inclusion, Héritage entre classes

- Créer une sous-classe SC d'une classe C, revient à définir un sous-type, au sens du polymorphisme d'inclusion du type défini par C. Ainsi, SC hérite de C et toute instance de SC possède les caractéristiques définies par C.




Figure (6) – Variétés de polymorphismes (extrait de Luca Cardelli, Peter Wegner : “On Understanding Types, Data Abstraction, and Polymorphism”. *ACM Comput. Surv.* 17(4) : 471-522 (1985))

- Héritage simple : toute méthode définie sur une surclasse est applicable (via envoi de message) aux instances des sous-classes.
- **Règle de redéfinition/spécialisation** : toute méthode de nom *m* d'une sous-classe SC d'une classe *C* définissant *m* en est une redéfinition.
- de plus :
 - pas de règles de covariante/contravariance, pas de *downcast*,
 - pas de surcharge,
 - polymorphisme paramétrique universel mais sans contrôle du compilateur,
- avantages et défauts à corréler à la pratique de tests systématiques et élaborés ? (Plus globalement voir *Agile Development*, ou la classe `TestCase` en *Pharo*).

3.8 Un exemple de hiérarchie : Les collections

```
1  ProtoObject #()
2      Object #()
3
4      Collection #()
5          Bag #('contents')
6              IdentityBag #()
7          CharSet #('map')
8          SequenceableCollection #()
9              ArrayedCollection #()
10                 Array #()
11                     ActionSequence #()
12                     DependentsArray #()
13                     WeakActionSequence #()
14                     WeakArray #()
15                     Array2D #('width' 'contents')
16                     B3DPrimitiveVertexArray #()
17                     Bitmap #()
18                     Heap #('array' 'tally' 'sortBlock')
19                     Interval #('start' 'stop' 'step')
20                         TextLineInterval #('internalSpaces' 'paddingWidth' 'lineHeight' 'baseline')
21                     LinkedList #('firstLink' 'lastLink')
22                         Semaphore #('excessSignals')
23                     MappedCollection #('domain' 'map')
24                     OrderedCollection #('array' 'firstIndex' 'lastIndex')
25                         GraphicSymbol #()
26                         SortedCollection #('sortBlock')
27                         UrlArgumentList #()
28                     SourceFileArray #()
29                         StandardSourceFileArray #('files')
30                     Set #('tally' 'array')
31                     Dictionary #()
32                         HtmlAttributes #()
33                         IdentityDictionary #()
34                             SystemDictionary #('cachedClassNames')
35                                 Environment #('envtName' 'outerEnvt')
36                                     SmalltalkEnvironment #()
37                         LiteralDictionary #()
38                         MethodDictionary #()
39                         PluggableDictionary #('hashBlock' 'equalBlock')
```

```
40         WeakKeyDictionary #()
43             ExternalFormRegistry #('lockFlag')
44             WeakIdentityKeyDictionary #()
45             WeakValueDictionary #()
46         IdentitySet #()
47         PluggableSet #('hashBlock' 'equalBlock')
48         WeakSet #('flag')
49     SkipList #('sortBlock' 'pointers' 'numElements' 'level' 'splice')
50         IdentitySkipList #()
51     WeakRegistry #('valueDictionary' 'accessLock')
```

3.9 Héritage et description différentielle




Figure (7) – *Hiérarchie d’héritage avec ajouts de fonctionnalités. Pour plus de détails sur la description différentielle, voir notes de cours “Réutilisation/Frameworks”.*

3.10 Schémas de spécialisation utilisant l'héritage

- Masquage : spécialisation d'une méthode sur une sous-classe.
- Masquage partiel : utilisation de la pseudo-variable `super`.
- Paramétrage par spécialisation (pseudo-variable `self`), ou par composition, classiques (voir cours "Réutilisation/Frameworks").

```
1 indexOf: anElement startingAt: start ifAbsent: exceptionBlock
2     "Answer the index of the first occurrence of anElement after
3     start
4     within the receiver. If the receiver does not contain
5     anElement,
6     answer the result of evaluating the argument, exceptionBlock."
7
8     start to: (self size)
9         do: [:index | (self at: index) = anElement ifTrue: [^ index]].
10    ^ exceptionBlock value
```

Listing (6) – Exemple sur la classe SequenceableCollection

3.11 Héritage statique classes-traits et traits-traits

Les *traits*^a sont des unités de réutilisation prédéfinies partageables et composables, intégrées par différents langages (Pharo, Php, Scals, ...).

Un *trait*, à distinguer d'une interface, définit des méthodes abstraites ou concrètes.

```
1 Trait named: #TComparable
2     uses: {}
3     package: 'Kernel-Traits'!
4
5 < aComparable
6     "Answer whether the receiver is less than the argument."
7     ^self subclassResponsibility
8
9 min: aComparable
10    "Answer the receiver or the argument, whichever is lesser."
11    self < aComparable
12        ifTrue: [^self]
13        ifFalse: [^aComparable]
14
15 ... etc
```

Listing (7) – Exemple de Trait

a. Nathanael Schärli, Stéphane Ducasse, Oscar Nierstrasz, Andrew P. Black : Traits : Composable Units of Behaviour. ECOOP 2003 : 248-274.

Les traits n'existent pas en Smalltalk originel, la mise en oeuvre présentée ici est celle de *Pharo*.

```

1  Object subclass: #CompteurWTrait
2      uses: TComparable
3      instanceVariableNames: 'valeur'
4      classVariableNames: ''

6  get
7      ^valeur

9  < aComparable
10     ^valeur < aComparable get

```

Listing (8) – Exemple d'utilisation d'un Trait

Toute classe utilisant (`use` relation) un trait `T` possède (héritage au moment de la définition) les méthodes définies par `T` :

- les méthodes de la classe utilisatrice (par exemple la méthode `=` de l'exemple) sont prioritaires par rapport à celles d'un trait utilisé (on peut donc redéfinir une méthode d'un trait),
- les méthodes d'un trait utilisé sont prioritaires par rapport à celles des superclasses de la classe utilisatrice.
- Une méthode définie sur un trait `T` n'a pas accès aux variables d'instances d'une classe qui utilise `T`.

Une classe peut utiliser plusieurs traits ; sous réserve de gérer (statiquement) les conflits de noms.

Un trait peut utiliser un autre trait.

Un trait peut être utilisé par un(e) ou plusieurs classes ou traits.

Un trait définir des attributs (résultat récent - *stateful traits*^a)

a. Pablo Tesone, Stéphane Ducasse, Guillermo Polito, Luc Fabresse, Noury Bouraqadi, **A new modular implementation for stateful traits**, Science of Computer Programming, Volume 195, 2020, 102470, ISSN 0167-6423.

```

1 Trait named: #TSortable
2   uses: {}
3   category: 'Collections-Abstract'
4
5 !TSortable methodsFor: 'sorting' !
6 sort
7     "Sort this collection into ascending order using the '<=' operator."
8     self sort: [:a :b | a <= b]
9     "-----"
10 Collection subclass: #SequenceableCollection
11   uses: TSortable
12   instanceVariableNames: ''
13   classVariableNames: ''
14   package: 'Collections-Abstract'

```

Listing (9) – Traits, exemple no 2. En utilisant le trait TSortable, la classe SequenceableCollection possède la méthode sort, toutes ses instances peuvent l'utiliser.

```

1 c := OrderedCollection new.
2 c add: 1; add: 19; add: 8; add: 11.
3 c sort -> an OrderedCollection(1 8 11 19)
4 c isKindOf: TSortable -> false

```

Listing (10) – Utilisation de la méthode sort du trait TSortable pour une instance de OrderedCollection, sous-classe de SequenceableCollection.

4 Le tout-objet et la méta-programmation

4.1 Définitions

Réflexivité : Capacité qu'a un système à donner à ses utilisateurs une représentation de lui-même en connexion causale avec sa représentation effective en machine

méta-entité : entité constitutive du méta-niveau donc du langage ou de son environnement

Entité de première classe : entité, possiblement une méta-entité, ayant une représentation accessible dans un programme, que l'on peut référencer, manipuler et inspecter, et éventuellement modifier.

En Smalltalk, les classes, les méthodes compilées, les méthodes, le source des méthodes, certaines structures de contrôle, éventuellement la pile d'exécution, l'environnement de programmation, le compilateur, ... sont des entités de première classe.

méta-programmation : programmation des entités constitutives du système (méta-entités) en utilisant le système.

4.2 Méta-programmation basique

4.2.1 Les classes comme “rvalues” standards

La “généricité paramétrique” est la capacité à contraindre les structures de données composites, comme les collections par exemple, à ne contenir que des éléments appartenant à des types identifiées dans le texte du programme.

En typage statique (voir les génériques *Java*, les templates *C++*) la vérification de telles contraintes peut être faite à la compilation.

Les classes comme “rvalues”, accompagnées d’une primitive de test de sous-typage, permettent en typage dynamique un contrôle dynamique de types.

Exemple :

```
1 Pile subclass: #PileTypee
2   instanceVariableNames: 'typeElements'
3   classVariableNames: ''
4   category: 'TP1'
6 push: element
7   (element isKindOf: typeElements)
8     ifTrue: [ super push: element ]
9     ifFalse: [ self error: 'Impossible d''empiler ', element printString ,
10              ' dans une pile de ', typeElements printString]
```

Listing (11) – test du type d'un objet stocké dans une variable. A ne pas confondre avec instanceOf de Java.style

4.2.2 Manipulation standard des objets primitifs (rock-bottom objects)

Chaque élément d'un type primitif (nombres, caractères, booléens, chaînes, tableaux) est représenté par un méta-objet qui permet son utilisation comme un objet standard. On peut en particulier lui envoyer des messages.

```
1 5 factorial "entier"  
2 'abcde' at: 3 "chaîne de caractère"  
3 #($a $b) reverse "tableau"  
4 true ifTrue: [#ofCourse] "booléen"  
5 false ifFalse: [#useless]  
6 nil isNil "null pointer or UndefinedObject"  
7 selecteur := #facto , #rial "concaténation de symboles"
```

Un type primitif est représenté par une classe mais son implantation n'est pas entièrement définie par cette classe et réside partiellement dans la machine virtuelle.

Il est possible de modifier les méthodes de ces classes (forcément dangereux) et d'en créer de nouvelles (exemple, la méthode `factorial` sur la classe `Integer`).

La hiérarchie des classes représentant les nombres.

```
1  Magnitude
2      Number ()
3          FixedPoint ('numerator' 'denominator' 'scale')
4          Fraction ('numerator' 'denominator')
5          Integer ()
6              LargeInteger ()
7                  LargeNegativeInteger ()
8                  LargePositiveInteger ()
9              SmallInteger ()
10         LimitedPrecisionReal ()
11             Double ()
12             Float ()
```

4.2.3 L'objet nil.

`nil` (`null` en *Java*) est la valeur par défaut du type référence, donc affectée à toute *l-value* (variable, case de tableau, etc) non explicitement initialisée.

Nil est l'unique instance (*Singleton*) de la classe `UndefinedObject` sur laquelle il est possible de définir des méthodes.

Il est donc possible d'envoyer un message à `nil`.

L'exception `nullPointerException` n'existe pas!

Le nom de la classe est certainement discutable.

```
nil isNil "true"
```

```
nil class "UndefinedObject"
```

Application à l'implantation des collections

La classe UndefinedObject offre une alternative originale pour l'implantation de certains types récurifs, par exemple List ou Arbre.

```
1 type List =  
2     Empty |  
3     Tuple of Object * List
```

dont certaines fonctions se définissent par une répartition sur les différents constructeurs algébriques :

```
1 length :: List -> Int  
2 lentgh Empty           = 0  
3 length Tuple val suite = 1 + (lentgh suite)
```

Mise en oeuvre (extrait 1)

```
Object subclass: #List
  instanceVariableNames: 'val suite'
  classVariableNames: ''
  poolDictionaries: ''
  category: 'Exemples'

!List methodsFor: 'accessing' !
suite
  ^suite

first
  ^val

first: element suite: uneListe
  val := element.
  suite := uneListe.
```

```
!List methodsFor: 'manipulating' !
```

```
addFirst: element
```

```
    ^List new first: element suite: self
```

```
length
```

```
    ^1 + suite length
```

```
append: aList
```

```
    ^(self suite append: aList) addFirst: self first
```

Mise en oeuvre (extrait 2)

```
!UndefinedObject methodsFor: 'ListManipulation'!
```

```
addFirst: element
```

```
    ^List with: element
```

```
length
```

```
    ^0
```

```
append: aList
```

```
    ^aList
```

4.2.4 Fermetures Lexicales - Addendum

Les structures de contrôle sont réalisées en Smalltalk par des méthodes définies sur les classes `Boolean` (conditionnelles, et, ou), `BlockClosure` (boucle “tantque”), `Integer` (boucle “for”), et utilisant des fermetures lexicales.

Une fermeture lexicale est une fonction anonyme créée dans l’environnement lexical de sa définition et exécutée dans cet environnement quel que soit son point d’activation.

Les fermetures en Smalltalk sont implantées en *Smalltalk* par la classe `BlockClosure`.

- Une fermeture est une constante littérale dont la valeur est elle-même.

```
[ :x | x + 1 ]  
= [ :x | x + 1 ]
```

- Une fermeture peut ainsi être utilisée comme *rvalue* (affectation, passage en argument) :

```
V := [ :x | x + 1 ]  
(date = '24/12') ifTrue: [ Transcript show: 'Noël' . ]
```

- Une fermeture est une fonction, que l’on peut donc invoquer. Comme c’est par ailleurs aussi un objet (tout est objet) on l’invoque en lui envoyant un message, ici le message `value`.

```
[ 2 + 3 ] value
```

```
= 5
```

- Si la fermeture possède des paramètres, on l'invoque en lui passant des arguments avec les messages `value:` , `value:value:`, etc (c'est moins beau qu'en *Scheme* mais c'est conforme au paradigme).

```
V value: 23
```

```
=24
```

```
[:i | i + 1] value: 33
```

```
= 34
```

4.2.5 Fermetures et retour de méthode

Une fermeture capture le point de retour du *block* ou elle est créée.

```
1  "une méthode de classe définie sur la classe Object  
2  qui crée une fermeture et la passe en argument à la méthode test2"  
3  test2Main  
4      self test2: [^1 + 2].  
5      Transcript show: '33333333'; cr.  
  
7  "la méthode de classe test2 définie également sur Object  
8  invoque la fermeture"  
9  test2: aBlock  
10     Transcript show: '11111111'; cr.  
11     aBlock value.  
12     Transcript show: '22222222'; cr.  
  
14 "exécution de Object test2Main"  
15 Object test2Main  
16 11111111  
17 =3
```

A condition que le point de retour existe toujours lors de l'exécution du *block*.

```
1 test1Main
2   | b |
3   b := self test1.
4   Transcript show: '11111111'; cr.
5   Transcript show: b value.
6   Transcript show: '22222222'; cr.
8 test1
9   ^[^1 + 2]
```

- The receiver tried to return result to homeContext that no longer exists.

4.2.6 Fermeture et Structures de contrôles

De par leur auto-évaluation, les fermetures lexicales permettent la définition de structures de contrôle.

```
1  "une méthode abstraite sur la classe Boolean"
3  ifTrue: alternativeBlock
4      "If the receiver is false (i.e., the condition is false), then the value is the
5      false alternative, which is nil. Otherwise answer the result of evaluating
6      the argument, alternativeBlock. Create an error notification if the
7      receiver is nonBoolean. Execution does not actually reach here because
8      the expression is compiled in-line."
10 self subclassResponsibility
```

4.3 Les Méta-classes en Smalltalk

4.3.1 Rappel : la solution Objvlisp




Figure (8) – Une classe et une méta-classe sont structurellement et fonctionnellement identiques

- Object est la racine de l'arbre d'héritage,

- **Class** est la racine de l'arbre d'instantiation, elle est instance d'elle-même et sous-classe de **Object**.

On parle à son propos d'un système à méta-classes explicites car le créateur d'une classe peut choisir la méta-classe qu'il souhaite instancier.

L'intérêt de cette solution est de permettre le découplage entre la superclasse de la méta-classe et la méta-classe de la super-classe. Une classe peut ainsi et par exemple être abstraite (instance de la méta-classe *AbstractClass*) sans que ses sous-classes le soient.

L'inconvénient de cette solution est qu'elle peut poser des problèmes de compatibilité (voir section [4.3.8](#)). Le langage *CLOS* qui réalise un sur-ensemble de la solution *Objvlistp* laisse au développeur le soin de résoudre ces problèmes éventuels de compatibilité via la définition de fonctions ad.hoc.

4.3.2 Motivations de la solution Smalltalk

La solution Smalltalk a été conçue avec deux objectifs ^a :

- Donner la possibilité de définir des méthodes sur les métaclasse tout en rendant les métaclasse invisibles au développeur qui n'a pas envie de les voir. L'invisibilité des méta-classes pour le développeur standard est réalisé par l'éditeur de programmes également nommé *browser*.
- Rendre impossible la création de métaclasse incompatibles.

a. Il y a donc deux façons de considérer ce système, la première est de l'utiliser sans chercher à entrer dans le détail de sa réalisation et d'apprécier son efficacité. Le bouton "class" du browser est dédié à la programmation des méthodes de classe et il n'est pas nécessaire de comprendre la mécanique interne pour l'utiliser. La seconde est d'aller y voir de plus près pour le plaisir de comprendre, ce qui prend tout son sens dans le cadre d'un cours sur la réflexivité.

4.3.3 Architecture du système de métaclasses




Figure (9) – *Classes et Métaclasses : Hiérarchies d’héritage et d’instantiation.*
 (figure : Gabriel Pavillet)




Figure (10) – *Classes et Métaclases : couches d’instanciation.* (figure : Gabriel Pavillet)

Les points clé :

- Toute classe `C` possède une unique métaclasse référençable via l'expression `C class`.
- Les deux hiérarchies d'héritage entre classes et d'héritage entre métaclasses sont isomorphes.
- `Object class` est la racine de la sous-hiérarchie des métaclasses.
- Le graphe d'héritage étant un arbre avec donc une racine unique (`Object`), le lien entre les hiérarchies de classes et de métaclasses est fait au niveau de la métaclasse `Object class` qui hérite de la classe `Class`.

Ceci se comprend, selon la sémantique usuelle^a du lien d'héritage, ainsi : une `Object class`, par exemple `Object`), est une sorte de classe.

a. AKO : a kind of, est une sorte de ... une voiture est une sorte de véhicule.

- Il y a 4 niveaux (conceptuels) d’instanciation. Un objet (couche 1) (par exemple un point), instances d’une classe (dans cet exemple `Point`), instance d’une métaclasse (dans cet exemple `Point class`), instance de l’unique méta-méta-classe (`Metaclass`). Le problème de regression infinie dans les descriptions est réglé par le fait que `metaclass` est instance de `metaclass class`, qui est une métaclasse standard de la couche 3, instance de `Metaclass`. `Metaclass class` a exactement le même format que toutes les autres métaclasses.

4.3.4 Structure des classes et des métaclasses (cf. fig. 11)

```

1 ProtoObject #()
2   Object #()
3     Behavior #('superclass' 'methodDict' 'format')
4       ClassDescription #('instanceVariables' 'organization')
5         Class #('subclasses' 'name' 'classPool' 'sharedPools'
6           'environment' 'category')
7           [ ... all the Metaclasses ... ]
           Metaclass #('thisClass')
```

Figure (11) – Les classes “Class” et “Metaclass” (et leurs attributs) définissant les classes et les méta-classes. Elles ont en commun tout ce qui est hérité de “ClassDescription”.

La classe `Class` déclare, en fermant transitivement la relation “est-sous-classe-de”, les attributs : “`superclass methodDict format instanceVariables organization subclasses name classPool environment category sharedPools`”. Toute sorte de `Class`, pourra posséder une valeur pour chacun d’eux.

Par exemple la valeur de l’attribut `instanceVariables` de la classe `Point` (instance de `Point class` qui hérite de `Class`) est ‘`x y z`’. Les variables de classes sont rangées dans l’attribut `classPool`.

La classe `Metaclass` déclare les attributs : “superclass methodDict format instanceVariables organization thisClass”.

`thisClass` (nom peu évocateur) référence la classe, unique instance (schéma *Singleton*) d’une métaclasse dans un contexte donné.

`thisClass` est utilisable dans les méthodes d’instance de la classe `Metaclass` qui n’a aucune sous-classe^a.

a. Exercice : créer une sous-classe MC de la classe `Metaclass`, qui définirait un nouveau type généraliste de meta-classes. La difficulté principale n’est pas dans la création de cette classe mais, comme les meta-classes sont créés automatiquement par le système, de modifier le système de création des classes pour qu’une classe qui en relève ne soit pas une instance de `Metaclass` mais de MC.




Figure (12) – *l'Inspection d'une classe montre que ses champs sont conformes à la déclaration des attributs réalisés dans la classe Class.*

4.3.5 Variable de classe versus variable d'instance de métaclasse

- **Rappel** : une variable de classe (*static en Java*) définie sur une classe **C** détient une valeur partagée par toutes les instances (au sens large) de **C** ; elle est accessible dans les méthodes d'instances et dans les méthodes de classes de **C** (et de ses sous-classes).
- **Nouveauté** : une variable d'instance^a **I** définie sur une classe **C** permet à chaque instance de **C** de détenir une valeur propre pour **I** ;
Si **C** est une métaclasse, on dira que **I** est une **variable d'instance de métaclasse** ; toute instance de **C** est une classe et possède une valeur pour **I**.

a. On rappelle que le terme “variable d'instance” dénote en Smalltalk ce qui est usuellement nommé “attribut”.

4.3.6 Utilisation standard des méta-classes : 1) création d'instances initialisées

Une classe complète avec ses méthodes de création d'instances : la classe `Pile`.

```
1 Object subclass: #Pile
2     instanceVariableNames: 'index buffer '
3     classVariableNames: ''
4     poolDictionaries: ''
5     category: 'Exercices'
6 ...
```

```

1  "-----"
2  Pile class
3      instanceVariableNames: ''
4
5  !Pile class methodsFor: 'creation'!
6  example
7      "self example"
8      "Rend une pile d'entiers de taille 10, pleine."
9      | aStack |
10     aStack := self new: 10.
11     1 to: aStack size do: [:i | aStack push: i].
12     ^aStack
13
14  new
15     "On fixe la taille par défaut à 10"
16     ^super new initialize: 10
17
18  new: taille
19     ^super new initialize: taille
20
21  taille: taille
22     ^self new: taille

```

4.3.7 Utilisation standard des méta-classes : 2) création de nouvelles méta-classes

Si l'on souhaite faire d'une classe une "memo classe" afin qu'elle mémorise la liste de ses instances, la solution naturelle est d'utiliser une variable d'instance de métaclasse.

```
1  EtudiantMemo class
2      instanceVariableNames: 'tous'

4  initialize
5  tous := OrderedCollection new.

7  new
8      | i |
9      i := super new.
10     tous add: i.
11     ^i

13  tous
14     ^tous
```

4.3.8 Avantage du modèle Smalltalk : la compatibilité des métaclases

La solution Smalltalk ne pose pas de problèmes de compatibilité.

Les problèmes de compatibilité surviennent avec les architectures à la `Objvlisp` lors de phase d'abstraction (passage au niveau meta - compatibilité ascendante) ou de la phase de réification (passage au niveau réifié - compatibilité descendante).

Problème de compatibilité Ascendante

Soient :

- une classe A, définissant une méthode `foo` : `'self class bar'`
- une classe B sous-classe de A
- la classe de A et sa méthode `bar` : `'...'`
- la classe de B, non sous-classe de la classe de A, et ne définissant pas de méthode `bar`
- une instance b de B

alors : `'b foo'` lève une l'exception "B ne comprend pas le message bar".

Cette situation est impossible avec le modèle *Smalltalk* où la classe de B ne peut pas ne pas être une sous-classe de la classe de A et hérite donc de la méthode `bar`.

Problème de Compatibilité Descendante

Soient :

- une méta-classe MA définissant une méthode `bar` : `'self new foo'`
- une classe A instance de MA définissant une méthode `foo`.
- une méta-classe MB, sous classe de MA
- une classe B instance de MB

'B bar' peut lever une exception si aucune méthode `foo` n'est définie dans une superclasse de B, en particulier si B n'est pas sous-classe de A.

4.3.9 Inconvénient du modèle Smalltalk

Il est impossible de découpler les hiérarchies de classes et de méta-classes. Ce qui pose des problèmes d'expression.

Par exemple si on crée une méta-classe `AbstractClass` dont les instances ne peuvent être instanciées (par une redéfinition appropriée de `new`). Alors toutes les sous-classes d'une instance de `AbstractClass` hériteront de cette redéfinition de `new`. Or, les classes abstraites ont bien sûr des sous-classes concrètes.

4.4 Les Méta-objets permettant d'accéder à la pile d'exécution

Smalltalk permet de manipuler chaque bloc (frame) de la pile d'exécution comme un objet de première classe avec une politique de "si-besoin" car la réification est une opération coûteuse.

Par exemple, le code suivant implante les structures de contrôle `catch` et `throw` permettant de réaliser des échappements à la Lisp (réalisation de branchements non locaux), que l'on peut voir comme les structures de contrôle de base nécessaires à l'implantation de mécanismes de gestion des exceptions.

`catch` permet de définir un point de reprise à n'importe quel point d'un programme et `returnToCatchWith:` interrompt l'exécution standard et la fait reprendre à l'instruction qui suit le `catch` correspondant (le symbole receveur détermine la correspondance).

Exemple d'application : interruption d'une recherche dès que l'on a trouvé l'élément recherché pour revenir à un point antérieur de l'exécution du programme.

4.4.1 Version Visualworks7

```
1 !Symbol methodsFor: 'catch-throw'!  
  
3 catch: aBlock  
4     "execute aBlock with a throw possibility"  
5     aBlock value.  
  
7 returnToCatchWith: aValue  
8     "Look down the stack for a catch, the mark of which is self,  
9     when found, transfer control (non local branch)."  
10    | catchMethod currentContext |  
11    currentContext := thisContext.  
12    catchMethod := Symbol compiledMethodAt: #catch:.  
13    [currentContext method == catchMethod and: [currentContext receiver == self]]  
14        whileFalse: [currentContext := currentContext sender].  
15    thisContext sender: currentContext sender.  
16    ^aValue
```

Listing (12) – version Visualworks-Smalltalk

4.4.2 Version Pharo6

```
1 !Symbol methodsFor: 'catch-throw'!  
  
3 catch: aBlock  
4     "execute aBlock with a throw possibility"  
5     aBlock value.  
  
7 returnToCatchWith: aValue  
8     | catchMethod currentContext |  
9     currentContext := thisContext.  
10    catchMethod := Symbol compiledMethodAt: #catch:.  
11    [currentContext method == catchMethod and: [currentContext receiver == self]]  
12        whileFalse: [currentContext := currentContext sender].  
13    currentContext return: aValue.  
14    ^aValue
```

Listing (13) – version Pharo-6

Exemple d'utilisation :

```
1 ^#Essai catch: [  
2     Transcript show: 'a';cr.  
3     Transcript show: 'b';cr.  
4     Transcript show: 'c';cr.  
5     #Essai returnToCatchWith: 22.  
6     Transcript show: 'd';cr.  
7     33]
```

Listing (14) – should display a b c (not d) in the Transcript and return 22 (not 33)

4.5 Méta-objets pour accéder au compilateur et aux méthodes compilées

Le compilateur Smalltalk est écrit en Smalltalk et est un objet de première classe. Cela permet de définir dynamiquement de nouvelles classes et méthodes.

L'exemple suivant est extrait de la réalisation d'un mini-tableur en Smalltalk. Dans cet exemple, les formules associées aux cellules sont représentées par des méthodes définies dynamiquement par le programme, lexicalement analysées et compilées à la volée.

Pour savoir comment ajouter, une fois compilée, une méthode à une classe, il faut étudier les protocoles définis sur les classes `Behavior` et `ClassDescription`.

```
1 Model subclass: #Cellule
2     instanceVariableNames: 'value formula internalFormula dependsFrom '
3     classVariableNames: ''
4     poolDictionaries: ''
5     category: 'Tableur'!
```

```

1 !Cellule methodsFor: 'compile formula'!
2 compileFormula: s
3     "Analyse lexicale, puis syntaxique puis generation de code pour la formule s"
4     | tokens newDep interne methodNode |
5     tokens := Scanner new scanTokens: s.
6     newDep := (tokens select: [:i | self isCaseReference: i]) asSet.
7     interne := 'execFormula\ | '.
8     newDep do: [:each | interne := interne , each , ' '].
9     interne := interne , '|\' '.
10    newDep do: [:each | interne := interne , each , ' := (Tableur current at: #' , each , ')
11                value.\ '].
12    interne := (interne , '^' , s) withCRs asText.
13    methodNode := UndefinedObject compilerClass new
14                compile: interne
15                in: UndefinedObject
16                notifying: nil
17                ifFail: [].
18    internalFormula := methodNode generate.
19    ^newDep

```

```
1 !Cellule methodsFor: 'exec formula'!  
2 executeFormula  
3     formula isNil  
4         ifFalse:  
5             [UndefinedObject addSelector: #execFormula withMethod:  
6                 internalFormula.  
7                 ^nil execFormula]  
8         ifTrue: [self error]  
9  
10 update: symbol  
11     symbol == #value ifTrue: [self setValue: self executeFormula]
```

Remerciements

Merci à René Paul Mages pour ses relectures attentives et remarques pertinentes. <https://renemages.wordpress.com/smalltalk/pharo/>